Exercise 2.6 Cummutation of ∇ on a scalar

```
{a,b,c,d,e,f,g,h,i,j,k,l,m,n,o,p,q,r,s,t,u#}::Indices(position=independent).

\nabla{#}::Derivative.
\partial{#}::PartialDerivative.

# covariant derivative of \phi

dphi := \nabla_{a}{\phi} -> \partial_{a}{\phi}.

# rules to hide and reveal \partial\phi

hide   := \partial_{a}{\phi} -> w_{a}.
reveal := w_{a} -> \partial_{a}{\phi}.

# template for covariant derivative of a dual-vector

deriv := \nabla_{a}{A?_{b}} -> \partial_{a}{A?_{b}} - \Gamma^{c}_{b a} A?_{c}.

# create an object

expr :=   \nabla_{a}{\nabla_{b}{\phi}}
        - \nabla_{b}{\nabla_{a}{\phi}}.   # cdb (ex-0206.101,expr)

# apply the rules, then simplify

substitute     (expr,dphi)                # cdb (ex-0206.102,expr)
substitute     (expr,hide)                # cdb (ex-0206.103,expr)
substitute     (expr,deriv)               # cdb (ex-0206.104,expr)
substitute     (expr,reveal)              # cdb (ex-0206.105,expr)
canonicalise   (expr)                     # cdb (ex-0206.106,expr)
```

$$
\begin{aligned}
\nabla_a\left(\nabla_b\phi\right)-\nabla_b\left(\nabla_a\phi\right) &= \nabla_a\left(\partial_b\phi\right)-\nabla_b\left(\partial_a\phi\right) && \texttt{(ex-0206.102)}\\
&= \nabla_a w_b-\nabla_b w_a && \texttt{(ex-0206.103)}\\
&= \partial_a w_b-\Gamma^c{}_{ba}w_c-\partial_b w_a+\Gamma^c{}_{ab}w_c && \texttt{(ex-0206.104)}\\
&= \partial_{ab}\phi-\Gamma^c{}_{ba}\partial_c\phi-\partial_{ba}\phi+\Gamma^c{}_{ab}\partial_c\phi && \texttt{(ex-0206.105)}\\
&= -\Gamma^c{}_{ba}\partial_c\phi+\Gamma^c{}_{ab}\partial_c\phi && \texttt{(ex-0206.106)}
\end{aligned}
$$